Problem 2.9

Problem 2.9. Use a computer to reproduce the table and graph in Figure 2.4: two Einstein solids, each containing three harmonic oscillators, with a total of six units of energy. Then modify the table and graph to show the case where one Einstein solid contains six harmonic oscillators and the other contains four harmonic oscillators (with the total number of energy units still equal to six). Assuming that all microstates are equally likely, what is the most probable macrostate, and what is its probability? What is the least probable macrostate, and what is its probability?

Einstein solid with 'n' oscillators and 'q' energy units:

In[∘]:=
```
Ω[n_, q_] := (q + n - 1)! / (q! (n - 1)!)
```

$$\Omega[n\_, q\_] := \frac{(q+n-1)!}{q!\,(n-1)!}$$

Reproduce the table in Figure 2.4.

Systems A and B

In[105]:=
```
nA = 3; nB = 3; qTotal = 6;
```

Systems A and B. The total number of microstates can be determined by considering all the ways to allocate the qTotal units of energy among the total nA+nB number of oscillators.

In[106]:=
```
ntotal = Ω[nA + nB, qTotal]
```

Out[106]=
```
462
```

Form a data table like those in the text, but add in a probability column for completeness.

In[107]:=
```
data = Table[{qA, ΩA = Ω[nA, qA], qB = qTotal - qA,
    ΩB = Ω[nB, qB], ΩA * ΩB, ΩA*ΩB / ntotal }, {qA, 0, qTotal}];
```

In[108]:=

```
TableForm[data,
 TableHeadings → {None, {"qA", "ΩA", "qB", "ΩB", "Ωtotal", "probability"}}]
```

Out[108]//TableForm=

qA	ΩA	qB	ΩB	Ωtotal	probability
0	1	6	28	28	$\frac{2}{33}$
1	3	5	21	63	$\frac{3}{22}$
2	6	4	15	90	$\frac{15}{77}$
3	10	3	10	100	$\frac{50}{231}$
4	15	2	6	90	$\frac{15}{77}$
5	21	1	3	63	$\frac{3}{22}$
6	28	0	1	28	$\frac{2}{33}$

The total number of microstates was computed above . Alternatively, you could add up column 5 to get the total number of microstates .

In[109]:=

```
Total[data〚All, 5〛] (* Add up column 5 to get the total number of macrostates *)
```

Out[109]=

462

In[110]:=

```
BarChart[data〚All, 5〛, LabelStyle → Larger, AxesLabel → {"qA", "Ωtotal"},
 ImageSize → Scaled[0.8], ChartLabels → data〚All, 1〛]
```

Out[110]=



Modify for nA = 6, nB = 4, qtotal still = 6.

In[111]:=
```
nA = 6; nB = 4; qTotal = 6;
```

In[112]:=
```
ntotal = Ω[nA + nB, qTotal]
```

Out[112]=
```
5005
```

Here, it will be convenient to express the probability as a real number, rather than an exact integer fraction.

In[113]:=
```
data = Table[{qA, ΩA = Ω[nA, qA], qB = qTotal - qA,
    ΩB = Ω[nB, qB], ΩA * ΩB, N[ΩA * ΩB / ntotal] }, {qA, 0, qTotal}];
```

In[114]:=
```
TableForm[data,
 TableHeadings → {None, {"qA", "ΩA", "qB", "ΩB", "Ωtotal", "probability"}}]
```

Out[114]//TableForm=

qA	ΩA	qB	ΩB	Ωtotal	probability
0	1	6	84	84	0.0167832
1	6	5	56	336	0.0671329
2	21	4	35	735	0.146853
3	56	3	20	1120	0.223776
4	126	2	10	1260	0.251748
5	252	1	4	1008	0.201399
6	462	0	1	462	0.0923077

In[115]:=

```
BarChart[data[[All, 5]], LabelStyle → Larger, AxesLabel → {"qA", "Ωtotal"},
 ImageSize → Scaled[0.8], ChartLabels → data[[All, 1]]]
```

Out[115]=



Most-probable macrostate

In[116]:=

```
TableForm[data,
 TableHeadings → {None, {"qA", "ΩA", "qB", "ΩB", "Ωtotal", "Probability"}}]
```

Out[116]//TableForm=

qA	ΩA	qB	ΩB	Ωtotal	Probability
0	1	6	84	84	0.0167832
1	6	5	56	336	0.0671329
2	21	4	35	735	0.146853
3	56	3	20	1120	0.223776
4	126	2	10	1260	0.251748
5	252	1	4	1008	0.201399
6	462	0	1	462	0.0923077

The most probable state is qA = 4, with 1260 total microstates, and a probability of about 25%.

In[117]:=

```
max = MaximalBy[data, #[[5]] &][[1]]
```

Out[117]=

```
{4, 126, 2, 10, 1260, 0.251748}
```

Least-probable macrostate

The least probable state is qA = 0, with 84 total microstates.

In[118]:=

```
min = MinimalBy[data, #[[5]] &][[1]]
```

Out[118]=

```
{0, 1, 6, 84, 84, 0.0167832}
```